Computing Fourier series with *Mathematica*

This notebook shows how to use *Mathematica* to automate the computation of partial sums of a Fourier series. We restrict ourselves to the case where the function has even symmetry about zero, so that the Fourier series has only cosine terms (you will have a homework problem in which you generalize this calculation to use both sines and cosines). A Fourier series can be defined on different intervals such as $[0, 2\pi]$ or $[-\pi, \pi]$. Here we use $[-\pi, \pi]$.

The Fourier series for an even function will be

$f(x) = \sum_{m=0}^{\infty} f_m \cos m x$

where the coefficients are given by the ratio of inner products

$f_m = \frac{\langle \cos m x,\, f(x) \rangle}{\langle \cos m x, \cos m x \rangle}.$

The inner product is $\langle f, g \rangle = \int_{-\pi}^{\pi} f(x)\, g(x)\, dx$.

In approximate calculations we do not take the sum to ∞, but truncate at some finite M :

$f(x) \approx \sum_{m=0}^{M} f_m \cos m x.$

Preliminary setup

Here we define some *Mathematica* functions we'll use to compute Fourier series: the basis functions, the inner product, the coefficients, and the M-th partial sum of the series.

- **Define the basis functions**

The even Fourier basis consists of the cosine functions

```
ϕ[m_, x_] = Cos[m x]
```

$\cos(m\, x)$

- **Define an inner product**

The inner product appropriate to Fourier analysis on periodic extensions of $[-\pi, \pi]$ is $\langle f, g \rangle = \int_{-\pi}^{\pi} f(x)\, g(x)\, dx$. (Technical *Mathematica* programming note: the inner product can't be evaluated yet, because the arguments f and g are dummy arguments, not yet defined. The ":=" used in the definition of fourierIP tells *Mathematica* to define fourierIP but to defer evaluation until the function fourierIP is actually used.)

```
fourierIP[f_, g_] := Integrate[f g, {x, -Pi, Pi}]
```

Here we show the form of the inner product:

```
fourierIP[u[x], v[x]]
```

$$\int_{-\pi}^{\pi} u(x)\, v(x)\, dx$$

Compute the expansion coefficients

With the inner product defined, we can compute the Fourier coefficients for a function "func". Note the use of deferred evaluation.

```
fourierCoeff[func_, m_] := fourierIP[func[x], ϕ[m, x]] / fourierIP[ϕ[m, x], ϕ[m, x]]
```

Show the general form of the Fourier coefficient:

```
fourierCoeff[ψ, m]
```

$$\frac{\int_{-\pi}^{\pi} \cos(m\, x)\, \psi(x)\, dx}{\frac{\sin(2\, m\, \pi)}{2\, m} + \pi}$$

Sum terms through order M

Here we define a function that computes the M-th partial sum of the Fourier series.

```
fourierSeries[M_, func_, x_] := Sum[fourierCoeff[func, m] Cos[m x], {m, 0, M}]
```

```
fourierSeries[M, ψ, x]
```

$$\sum_{m=0}^{M} \frac{\cos(m\, x) \int_{-\pi}^{\pi} \cos(m\, x)\, \psi(x)\, dx}{\frac{\sin(2\, m\, \pi)}{2\, m} + \pi}$$

At this point, we’re ready to go!

Example: Fourier series for a triangle wave

Define an expression for the function we want to expand in a Fourier series

We’ll approximate a triangle wave by a Fourier series. First we define an expression for one period of a triangle wave.

```
triangleWave[x_] = Piecewise[{{1 - Abs[x / Pi], Abs[x] ≤ Pi}}]
```

$$\begin{cases} 1 - \frac{|x|}{\pi} & |x| \leq \pi \end{cases}$$

Here’s what the function looks like.

```
Plot[triangleWave[x], {x, -4 Pi, 4 Pi}, PlotRange → All]
```



We’ll also want to see the periodic extension. The following function will replicate the triangle pulse over several periods.

```
periodicExtension[func_, nPeriods_] := Sum[func[x + 2 k Pi], {k, -nPeriods, nPeriods}]
```

Now we can plot a few periods of the triangle wave:

```
Plot[periodicExtension[triangleWave, 4], {x, -4 Pi, 4 Pi}, PlotRange → All]
```



- **Compute the fourier series for the triangle wave**

Let’s first look at the Fourier coefficients.

```
fourierCoeff[triangleWave, m]
```

$$-\frac{2(\cos(m\pi)-1)}{m^2\pi\left(\frac{\sin(2m\pi)}{2m}+\pi\right)}$$

Next, compute the 2nd, 4th, and 16th partial sums:

```
fs2[x_] = fourierSeries[2, triangleWave, x]
```

$$\frac{4\cos(x)}{\pi^2}+\frac{1}{2}$$

```
fs4[x_] = fourierSeries[4, triangleWave, x]
```

$$\frac{4\cos(x)}{\pi^2}+\frac{4\cos(3x)}{9\pi^2}+\frac{1}{2}$$

```
fs16[x_] = fourierSeries[16, triangleWave, x]
```

$$\frac{4\cos(x)}{\pi^2}+\frac{4\cos(3x)}{9\pi^2}+\frac{4\cos(5x)}{25\pi^2}+\frac{4\cos(7x)}{49\pi^2}+\frac{4\cos(9x)}{81\pi^2}+\frac{4\cos(11x)}{121\pi^2}+\frac{4\cos(13x)}{169\pi^2}+\frac{4\cos(15x)}{225\pi^2}+\frac{1}{2}$$

We can plot the partial sums against x.

```
Plot[{fs2[x], fs4[x], fs16[x]}, {x, -Pi, Pi}]
```



To see how the accuracy improves with increasing number of terms, plot the difference between the series and the exact triangle wave:

```
Plot[{fs2[x] - triangleWave[x],
  fs4[x] - triangleWave[x], fs16[x] - triangleWave[x]}, {x, -Pi, Pi}]
```



Finally, plot the Fourier series over several periods, to show that it automatically "builds in" the periodic extension of the square wave.

```
Plot[{fs2[x], fs16[x]}, {x, -6 Pi, 6 Pi}]
```